Dror Bar-Natan: Classes: 2014-15: Math 1100 Algebra I:

JCF Tricks and Programs

Row and Column Operations

Row operations are performed by left-multiplying N by some properly-positioned 2×2 matrix and at the same time left-multiplying the "tracking matrix" P by the same 2×2 matrix. Column operations are similar, with left replaced by right and P by Q.

```
RowOp[i_, j_, mat_] := Module[{TT = II},
   TT[[{i, j}, {i, j}]] = mat;
   NN = Simplify[TT.NN]; PP = Simplify[TT.PP]];
ColOp[i_, j_, mat_] := Module[{TT = II},
   TT[[{i, j}, {i, j}]] = mat;
   NN = Simplify[NN.TT]; QQ = Simplify[QQ.TT]];
```

Swapping Rows and Columns

```
SwapRows[i_, j_] := RowOp[i, j, (0 1
                                 1 0)];
SwapColumns[i_, j_] := ColOp[i, j, (0 1
                                    1 0)];
SwapBoth[i_, j_] := (SwapRows[i, j]; SwapColumns[i, j];)
```

The "GCD" Trick

If $q = \gcd(a, b) = s\,a + t\,b$, the equality $\begin{pmatrix} s & t \\ -b/q & a/q \end{pmatrix}\begin{pmatrix} a \\ b \end{pmatrix} = \begin{pmatrix} q \\ 0 \end{pmatrix}$ allows us to replace pairs of entries in the same column by their greatest common divisor (and a zero!), using invertible row operations. A similar trick works for rows.

```
? PolynomialExtendedGCD
```

PolynomialExtendedGCD[$poly_1$, $poly_2$, x] gives the extended GCD of $poly_1$ and $poly_2$ treated as univariate polynomials in x.
PolynomialExtendedGCD[$poly_1$, $poly_2$, x, Modulus → p] gives the extended GCD over the integers mod prime p. »

```
GCDTrick[{i_, j_}, k_] := Module[{a, b, q, s, t},
   {q, {s, t}} = PolynomialExtendedGCD[a = NN[[i, k]], b = NN[[j, k]], x];
   RowOp[i, j, (  s     t
                -b / q  a / q)]];
GCDTrick[k_, {i_, j_}] := Module[{a, b, q, s, t},
   {q, {s, t}} = PolynomialExtendedGCD[a = NN[[k, i]], b = NN[[k, j]], x];
   ColOp[i, j, (s  -b / q
                t   a / q)]];
```

Factoring Diagonal Entries

If $1 = \gcd(a, b) = s\,a + t\,b$, the equality $\begin{pmatrix} s\,a & 1 \\ -t\,b & 1 \end{pmatrix} \begin{pmatrix} 1 & 0 \\ 0 & a\,b \end{pmatrix} \begin{pmatrix} a & -b \\ t & s \end{pmatrix} = \begin{pmatrix} a & 0 \\ 0 & b \end{pmatrix}$ is an invertible row-column-operations proof of the isomorphism $\frac{R}{\langle a\rangle} \oplus \frac{R}{\langle b\rangle} \simeq \frac{R}{\langle ab\rangle}$.

```
SplitToSum[i_, j_, a_, b_] := Module[{q, s, t, T1, T2},
  {q, {s, t}} = PolynomialExtendedGCD[a, b, x];
  If[q == 1, RowOp[i, j, (s a   1
                           -t b  1)]; ColOp[i, j, (a  -b
                                                   t   s)]]];
```

The Jordan Trick

A repeated application of the identity $\begin{pmatrix} p^{k-1} & -1 \\ 1 & 0 \end{pmatrix} . \begin{pmatrix} 1 & 0 \\ 0 & p^k \end{pmatrix} . \begin{pmatrix} 1 & p \\ 0 & 1 \end{pmatrix} = \begin{pmatrix} p^{-1+k} & 0 \\ 1 & p \end{pmatrix}$ will bring a matrix like $\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & p^4 \end{pmatrix}$ to the “Jordan” form of $\begin{pmatrix} p & 0 & 0 & 0 \\ 1 & p & 0 & 0 \\ 0 & 1 & p & 0 \\ 0 & 0 & 1 & p \end{pmatrix}$, using invertible row and column operations.

```
JordanTrick[i_, j_, p_, s_] := (RowOp[i, j, (p^(s-1)  -1
                                              1       0)]; ColOp[i, j, (1  p
                                                                        0  1)]);
```

```
$Post = If[# =!= Null, #,
     Print[StringForm["`1` `5` `2` . `3` . `4`",
       NN // MatrixForm, PP // MatrixForm, MM // MatrixForm, QQ // MatrixForm,
       If[Expand[NN] === Expand[PP.MM.QQ], "=", "≠"]
      ]]
    ] &;
```

NN ≠ PP . MM . QQ

Running the JCF Programs

Matrix 1 - 3x3, 3 eigenvalues.

```
(n = 3; AA = (3   0   0
              4  -2  -6
             -2   0   1);

 PP = QQ = II = IdentityMatrix[n]; MM = x II - AA; NN = PP.MM.QQ;)
```

$$\begin{pmatrix} -3+x & 0 & 0 \\ -4 & 2+x & 6 \\ 2 & 0 & -1+x \end{pmatrix} = \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix} . \begin{pmatrix} -3+x & 0 & 0 \\ -4 & 2+x & 6 \\ 2 & 0 & -1+x \end{pmatrix} . \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

```
SwapRows[1, 2]
```

{{0, 1, 0}, {1, 0, 0}, {0, 0, 1}}

```
SwapColumns[2, 3]
```

{{1, 0, 0}, {0, 0, 1}, {0, 1, 0}}

```
GCDTrick[{1, 2}, 1]
```

$\left\{\left\{0, -\frac{1}{4}, 0\right\}, \{-4, 3-x, 0\}, \{0, 0, 1\}\right\}$

```
GCDTrick[1, {1, 2}]
```

$\left\{\left\{1, \frac{3}{2}, 0\right\}, \{0, 0, 1\}, \{0, 1, 0\}\right\}$

```
GCDTrick[1, {1, 3}]
```

$\left\{\left\{1, \frac{3}{2}, \frac{2+x}{4}\right\}, \{0, 0, 1\}, \{0, 1, 0\}\right\}$

```
GCDTrick[{1, 3}, 1]
```

$\left\{\left\{0, -\frac{1}{4}, 0\right\}, \{-4, 3-x, 0\}, \left\{0, \frac{1}{2}, 1\right\}\right\}$

```
GCDTrick[2, {2, 3}]
```

$\left\{\left\{1, -\frac{1}{4}, 0\right\}, \{0, 0, -6\}, \left\{0, -\frac{1}{6}, 2+x\right\}\right\}$

```
GCDTrick[{2, 3}, 2]
```

$\left\{\left\{0, -\frac{1}{4}, 0\right\}, \left\{\frac{4}{5}, \frac{1}{5}(-6+x), -\frac{6}{5}\right\}, \left\{-\frac{2}{3}(2+x), \frac{1}{6}\left(-3+4x-x^2\right), -3+x\right\}\right\}$

```
GCDTrick[2, {2, 3}]
```

$\left\{\left\{1, -\frac{1}{4}, -\frac{3}{10}\left(-2+x+x^2\right)\right\}, \{0, 0, -6\}, \left\{0, -\frac{1}{6}, \frac{1}{5}\left(12+4x-x^2\right)\right\}\right\}$

```
SplitToSum[1, 3, (-3 + x), (-2 + x + x^2)]
```

$\left\{\left\{-3+x-\frac{3}{100}\left(-2+x+x^2\right), -\frac{1}{4}, \frac{1}{100}\left(176-94x-85x^2+3x^3\right)\right\},\right.$
$\left.\left\{-\frac{3}{5}, 0, \frac{3(4+x)}{5}\right\}, \left\{\frac{1}{50}\left(12+4x-x^2\right), -\frac{1}{6}, \frac{1}{50}(4+x)\left(-12-4x+x^2\right)\right\}\right\}$

```
Factor[-2 + x + x^2]
```

$(-1+x)(2+x)$

```
SplitToSum[2, 3, (-1 + x), (2 + x)]
```

$$\left\{\left\{-3+x-\frac{3}{100}\left(-2+x+x^2\right),\ \frac{1}{300}\left(251-169x-85x^2+3x^3\right),\ \frac{1}{300}\left(-26+169x+85x^2-3x^3\right)\right\},\ \left\{-\frac{3}{5},\ \frac{4+x}{5},\ \frac{1}{5}(-4-x)\right\},\ \left\{\frac{1}{50}\left(12+4x-x^2\right),\ \frac{1}{150}\left(-23-53x+x^3\right),\ \frac{1}{150}\left(98+53x-x^3\right)\right\}\right\}$$

```
MatrixForm /@ {Coefficient[NN, x, 1], BB = -Coefficient[NN, x, 0]}
```

$$\left\{\begin{pmatrix}1&0&0\\0&1&0\\0&0&1\end{pmatrix},\ \begin{pmatrix}3&0&0\\0&1&0\\0&0&-2\end{pmatrix}\right\}$$

```
(CC = Sum[
    MatrixPower[BB, k].Coefficient[PP, x, k],
    {k, 0, n}
  ]) // MatrixForm
```

$$\begin{pmatrix}-\frac{10}{3}&0&0\\-2&0&-2\\0&-\frac{5}{2}&-5\end{pmatrix}$$

```
CC.AA.Inverse[CC] // MatrixForm
```

$$\begin{pmatrix}3&0&0\\0&1&0\\0&0&-2\end{pmatrix}$$

Matrix 2 - 3x3, one Jordan block.

$$\left(n = 3;\ AA = \begin{pmatrix}-\frac{5}{2}&-11&\frac{9}{2}\\-\frac{1}{2}&1&\frac{1}{2}\\-\frac{19}{2}&-16&\frac{21}{2}\end{pmatrix};\right.$$

$$\left. PP = QQ = II = \text{IdentityMatrix}[n];\ MM = x\,II - AA;\ NN = PP.MM.QQ;\right)$$

$$\begin{pmatrix}\frac{5}{2}+x&11&-\frac{9}{2}\\\frac{1}{2}&-1+x&-\frac{1}{2}\\\frac{19}{2}&16&-\frac{21}{2}+x\end{pmatrix} = \begin{pmatrix}1&0&0\\0&1&0\\0&0&1\end{pmatrix} . \begin{pmatrix}\frac{5}{2}+x&11&-\frac{9}{2}\\\frac{1}{2}&-1+x&-\frac{1}{2}\\\frac{19}{2}&16&-\frac{21}{2}+x\end{pmatrix} . \begin{pmatrix}1&0&0\\0&1&0\\0&0&1\end{pmatrix}$$

```
(GCDTrick[1, {1, 2}]; GCDTrick[1, {1, 3}]; GCDTrick[{1, 2}, 1];
 GCDTrick[{1, 3}, 1]; GCDTrick[2, {2, 3}]; GCDTrick[{2, 3}, 2];)
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & (-3+x)^3 \end{pmatrix} = \begin{pmatrix} 1 & 0 & 0 \\ \frac{1-x}{11} & 1 & 0 \\ \frac{1}{11}\left(89-166x+70x^2-9x^3\right) & 105-61x+9x^2 & 1 \end{pmatrix}$$

$$. \begin{pmatrix} \frac{5}{2}+x & 11 & -\frac{9}{2} \\ \frac{1}{2} & -1+x & -\frac{1}{2} \\ \frac{19}{2} & 16 & -\frac{21}{2}+x \end{pmatrix} . \begin{pmatrix} 0 & -\frac{81}{2} & -10+\frac{9x}{2} \\ \frac{1}{11} & -\frac{9}{2} & \frac{1}{2}(-2+x) \\ 0 & -\frac{67}{2}-9x & -8+\frac{3x}{2}+x^2 \end{pmatrix}$$

```
JordanTrick[2, 3, x - 3, 3]
```

$$\left\{\left\{0, -\frac{81}{2}, \frac{223}{2}-36x\right\}, \left\{\frac{1}{11}, -\frac{9}{2}, \frac{25}{2}-4x\right\}, \left\{0, -\frac{67}{2}-9x, \frac{185}{2}-5x-8x^2\right\}\right\}$$

```
JordanTrick[1, 2, x - 3, 2]
```

$$\left\{\left\{0, -\frac{81}{2}, \frac{223}{2}-36x\right\}, \left\{\frac{1}{11}, \frac{1}{22}(-105+2x), \frac{25}{2}-4x\right\}, \left\{0, -\frac{67}{2}-9x, \frac{185}{2}-5x-8x^2\right\}\right\}$$

```
MatrixForm /@ {Coefficient[NN, x, 1], BB = -Coefficient[NN, x, 0]}
```

$$\left\{\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}, \begin{pmatrix} 3 & 0 & 0 \\ -1 & 3 & 0 \\ 0 & -1 & 3 \end{pmatrix}\right\}$$

```
(CC = Sum[
    MatrixPower[BB, k].Coefficient[PP, x, k],
    {k, 0, n}
   ]) // MatrixForm
```

$$\begin{pmatrix} -2 & 3 & 1 \\ -1 & 7 & 0 \\ -1 & 9 & 0 \end{pmatrix}$$

```
CC.AA.Inverse[CC] // MatrixForm
```

$$\begin{pmatrix} 3 & 0 & 0 \\ -1 & 3 & 0 \\ 0 & -1 & 3 \end{pmatrix}$$

Matrix 3 - 4x4, mixed Jordan form.

$$\left(n = 4;\ AA = \begin{pmatrix} 1 & -2 & 0 & -2 \\ \frac{1}{4} & \frac{5}{2} & 0 & \frac{3}{2} \\ \frac{5}{2} & 5 & 2 & 3 \\ \frac{1}{4} & \frac{1}{2} & 0 & \frac{3}{2} \end{pmatrix};\right.$$

```
PP = QQ = II = IdentityMatrix[n]; MM = x II - AA; NN = PP.MM.QQ;
```

$$\begin{pmatrix} -1+x & 2 & 0 & 2 \\ -\frac{1}{4} & -\frac{5}{2}+x & 0 & -\frac{3}{2} \\ -\frac{5}{2} & -5 & -2+x & -3 \\ -\frac{1}{4} & -\frac{1}{2} & 0 & -\frac{3}{2}+x \end{pmatrix} = \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix} . \begin{pmatrix} -1+x & 2 & 0 & 2 \\ -\frac{1}{4} & -\frac{5}{2}+x & 0 & -\frac{3}{2} \\ -\frac{5}{2} & -5 & -2+x & -3 \\ -\frac{1}{4} & -\frac{1}{2} & 0 & -\frac{3}{2}+x \end{pmatrix} . \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

```
(GCDTrick[{1, 2}, 1]; GCDTrick[1, {1, 2}]; GCDTrick[1, {1, 4}]; GCDTrick[{1, 3}, 1];
 GCDTrick[{1, 4}, 1]; GCDTrick[2, {2, 3}]; GCDTrick[2, {2, 4}];
 GCDTrick[{2, 3}, 2]; GCDTrick[{2, 4}, 2]; GCDTrick[3, {3, 4}];)
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & -2+x & 0 \\ 0 & 0 & 0 & (-2+x)^2\,(-1+x) \end{pmatrix} = \begin{pmatrix} 0 & -4 & 0 & 0 \\ \frac{1}{4} & -1+x & 0 & 0 \\ -6+\frac{9x}{2} & 2\left(7-21x+9x^2\right) & 1 & 0 \\ \frac{1}{2}\left(-9+11x-3x^2\right) & 17-40x+28x^2-6x^3 & 0 & 1 \end{pmatrix}$$

$$. \begin{pmatrix} -1+x & 2 & 0 & 2 \\ -\frac{1}{4} & -\frac{5}{2}+x & 0 & -\frac{3}{2} \\ -\frac{5}{2} & -5 & -2+x & -3 \\ -\frac{1}{4} & -\frac{1}{2} & 0 & -\frac{3}{2}+x \end{pmatrix} . \begin{pmatrix} 1 & -12 & 0 & 2-2x \\ 0 & 9 & 0 & -2+\frac{3x}{2} \\ 0 & 0 & 1 & -2+3x \\ 0 & -13+6x & 0 & 3-\frac{7x}{2}+x^2 \end{pmatrix}$$

```
SwapBoth[1, 3]
```

$$\begin{pmatrix} -2+x & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & (-2+x)^2\,(-1+x) \end{pmatrix} = \begin{pmatrix} -6+\frac{9x}{2} & 2\left(7-21x+9x^2\right) & 1 & 0 \\ \frac{1}{4} & -1+x & 0 & 0 \\ 0 & -4 & 0 & 0 \\ \frac{1}{2}\left(-9+11x-3x^2\right) & 17-40x+28x^2-6x^3 & 0 & 1 \end{pmatrix}$$

$$. \begin{pmatrix} -1+x & 2 & 0 & 2 \\ -\frac{1}{4} & -\frac{5}{2}+x & 0 & -\frac{3}{2} \\ -\frac{5}{2} & -5 & -2+x & -3 \\ -\frac{1}{4} & -\frac{1}{2} & 0 & -\frac{3}{2}+x \end{pmatrix} . \begin{pmatrix} 0 & -12 & 1 & 2-2x \\ 0 & 9 & 0 & -2+\frac{3x}{2} \\ 1 & 0 & 0 & -2+3x \\ 0 & -13+6x & 0 & 3-\frac{7x}{2}+x^2 \end{pmatrix}$$

```
SplitToSum[2, 4, (-1 + x), (-2 + x)^2]
```

$$\left\{\{0, -14(-1+x), 1, 54-56x+14x^2\}, \left\{0, -11+\frac{21x}{2}, 0, \frac{1}{2}(-84+85x-21x^2)\right\}, \{1, -2+3x, 0, -(-3+x)(-2+3x)\}, \left\{0, 16-\frac{45x}{2}+7x^2, 0, 61-\frac{179x}{2}+\frac{87x^2}{2}-7x^3\right\}\right\}$$

```
SwapBoth[1, 2]
```

$$\begin{pmatrix} -1+x & 0 & 0 & 0 \\ 0 & -2+x & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & (-2+x)^2 \end{pmatrix} = \begin{pmatrix} \frac{1}{4}(-21+26x-7x^2) & 20-47x+33x^2-7x^3 & 0 & 1 \\ -6+\frac{9x}{2} & 2(7-21x+9x^2) & 1 & 0 \\ 0 & -4 & 0 & 0 \\ \frac{1}{4}(-22+26x-7x^2) & 21-48x+33x^2-7x^3 & 0 & 1 \end{pmatrix}$$

$$\cdot \begin{pmatrix} -1+x & 2 & 0 & 2 \\ -\frac{1}{4} & -\frac{5}{2}+x & 0 & -\frac{3}{2} \\ -\frac{5}{2} & -5 & -2+x & -3 \\ -\frac{1}{4} & -\frac{1}{2} & 0 & -\frac{3}{2}+x \end{pmatrix} \cdot \begin{pmatrix} -14(-1+x) & 0 & 1 & 54-56x+14x^2 \\ -11+\frac{21x}{2} & 0 & 0 & \frac{1}{2}(-84+85x-21x^2) \\ -2+3x & 1 & 0 & -(-3+x)(-2+3x) \\ 16-\frac{45x}{2}+7x^2 & 0 & 0 & 61-\frac{179x}{2}+\frac{87x^2}{2}-7x^3 \end{pmatrix}$$

```
JordanTrick[3, 4, x - 2, 2]
```

$$\left\{\{-14(-1+x), 0, 1, 52-55x+14x^2\}, \left\{-11+\frac{21x}{2}, 0, 0, \frac{1}{2}(-84+85x-21x^2)\right\}, \{-2+3x, 1, 0, -(-3+x)(-2+3x)\}, \left\{16-\frac{45x}{2}+7x^2, 0, 0, 61-\frac{179x}{2}+\frac{87x^2}{2}-7x^3\right\}\right\}$$

```
MatrixForm /@ {Coefficient[NN, x, 1], BB = -Coefficient[NN, x, 0]}
```

$$\left\{\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}, \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 2 & 0 & 0 \\ 0 & 0 & 2 & 0 \\ 0 & 0 & -1 & 2 \end{pmatrix}\right\}$$

```
(CC = Sum[
    MatrixPower[BB, k].Coefficient[PP, x, k],
    {k, 0, n}
   ]) // MatrixForm
```

$$\begin{pmatrix} -\frac{1}{2} & -1 & 0 & 1 \\ 3 & 2 & 1 & 0 \\ -\frac{1}{2} & -1 & 0 & -1 \\ -\frac{1}{2} & 0 & 0 & 0 \end{pmatrix}$$

```
CC.AA.Inverse[CC] // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 2 & 0 & 0 \\ 0 & 0 & 2 & 0 \\ 0 & 0 & -1 & 2 \end{pmatrix}$$